The Mathematica® Journal

The Arbelos

Incircle, Radical Circle, Radical Axis, Twins, Generalizations, and Proofs without Words

Jaime Rangel-Mondragón

This article systematically verifies a series of properties of an ancient figure called the arbelos. It includes some new discoveries and extensions contributed by the author.

Introduction

Motivated by the computational advantages offered by *Mathematica,* I decided some time ago to embark on collecting and implementing properties of the fascinating geometric figure called the arbelos. I have since been impressed by the large number of surprising discoveries and computational challenges that have sprung out of the growing literature concerning this remarkable object. I recall its resemblance to the lower part of the iconic canopied penny-farthing bicycle of the 1960s TV series *The Prisoner*, Punch's jester cap (of *Punch and Judy* fame), and a yin-yang symbol with one arc inverted; see Figure 1. There is now an online specialized catalog of Archimedean circles (circles contained in the arbelos) [1] and important applications outside the realm of mathematics and computer science [2] of arbelos-related properties.

Many famous names are involved in this fascinating theme, among them Archimedes (killed by a Roman soldier in 212 BC), Pappus (320 AD), Christian O. Mohr (1835–1918), Victor Thébault (1882–1960), Leon Bankoff (1908–1997), and Martin Gardner (1914–2010). Recently, they have been succeeded by Clayton Dodge, Peter Y. Woo, Thomas Schoch, Hiroshi Okumura, and Masayuki Watanabe, among others.

Leon Bankoff was the person who stimulated the extraordinary attention on the arbelos over the last 30 years. Schoch drew Bankoff's attention to the arbelos in 1979 by discovering several new Archimedean circles. He sent a 20-page handwritten note to Martin Gardner, who forwarded it to Bankoff, who then gave a 10-chapter manuscript copy to Dodge in 1996. Due to Bankoff's death, a planned joint work was interrupted until Dodge reported some discoveries [3]. In 1999 Dodge said that it would take him five to ten years to sort all the material in his possession, then filling three suitcases. Currently this work is still forthcoming. Not surprisingly, like Volume 4 of *The Art of Computer Programming*, it appears that important work needs a substantial time to be developed.

▲ **Figure 1.** *The Prisoner*'s penny-farthing bicycle, Punch and Judy, a physical arbelos.

The arbelos ("shoemaker's knife" in Greek) is named for its resemblance to the blade of a knife used by cobblers (Figure 1). The arbelos is a plane region bounded by three semicircles sharing a common baseline (Figure 2). Archimedes appears to have been the first to study its mathematical properties, which he included in propositions 4 through 8 of his *Liber assumptorum* (or *Book of Lemmas*). This work might not be entirely by Archimedes, as was recently revealed through an Arabic translation of the *Book of Lemmas* that mentions Archimedes repeatedly without fully recognizing his authorship (some even believe this work to be spurious [4]). The *Book of Lemmas* also contained Archimedes's famous *Problema Bovinum* [5].

This article aims at systematically enumerating selected properties of the arbelos, without attempting to be exhaustive. Our purpose is to develop a uniform computational methodology in order to tackle those properties in a pedagogical setting. A sequence of properties is arranged and subsequently verified by testing the computationally equivalent predicates. This work includes some discoveries and extensions contributed by the author.

We refer to the largest semicircle as the *top arc* and the two small ones as the left and right *side arcs,* or just the *side arcs* when there is no need to distinguish them. We use a and b to denote their respective radii (the top arc thus has radius $a + b$). A *segment* between two points is an undirected line segment going from one point to the other, while a *line* through two points is the infinite straight line through the two points. A traditional abuse of notation uses AB for both the line segment joining the points A and B and the length of the segment, depending on the context; modern usage is to write $|AB|$ for the length of the segment.

This function displays the arbelos.

```
arbelos[a_, b_] :=
 {
  Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
  White, Disk[{a, 0}, a, {0, π}],
  Disk[{2 a + b, 0}, b, {0, π}],
  Black, Circle[{a + b, 0}, a + b, {0, π}],
  Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
 }
```

This draws the basic arbelos.

```
Graphics[
 {
  arbelos[2, 1],
  {
   Gray, Disk[{2, 0}, .04], Disk[{5, 0}, .04],
   Arrow[{{2, 0}, {.7, 1.5}}], Arrow[{{5, 0}, {5.7, .7}}]
  }, Style[{Text["a", {1.2, .7}], Text["b", {5.5, .2}]},
   Italic, 12]
 }, PlotRange → {{-.1, 6.1}, {-.1, 3.1}},
 ImageSize → {425, 210}
]
```



▲ **Figure 2.** The arbelos.

Property 1

The perimeter of the arbelos is equal to the circumference of its largest circle.

In other words, the total length of the side arcs equals the length of the top arc. This property is related to an intriguing paradox [6].

Property 2

The area of the arbelos is equal to the area of the circle of diameter BR.

This was lemma 4 of the *Book of Lemmas* (see Figure 3) [7, 8].

These two properties are easily verified by simultaneously testing two equalities.

```
Module[{a, b},
 Simplify[(π (a + b) + π a + π b == π (2 (a + b))) ⋀
    (π (a + b)^2 / 2 - π a^2 / 2 - π b^2 / 2 == π (√(a b))^2), a > 0 ⋀ b > 0]
]
```

```
True
```

The function `drawpoints` is used to display specific points as red disks.

```
drawpoints[points_, s_: .05] :=
 {EdgeForm[Black], Red, Disk[#, s] & /@ points}
```

```
Graphics[{
  arbelos[2, 1],
  {Gray, Circle[{4, 1.41}, 1.41],
   Line[{{4, 0}, {4, 2.82}}]},
  Style[{Text["B", {4, 0}, {-3, -1}],
    Text["R", {4, 2.8}, {-3, -1}]}, 12],
  drawpoints[{{4, 0}, {4, 2.8}}]},
 PlotRange → {{-.1, 6.1}, {-.1, 3.1}}, ImageSize → {425, 210}]
```



▲ **Figure 3.** The area of the circle of diameter BR (the radical circle) is equal to the area of the arbelos.

■ The Radical Circle

The circle in Figure 3 is called the *radical circle* of the arbelos and the line BR is its *radical axis* (this terminology will be clarified in Generalizations). To illustrate properties 3–11 and 25, 26, we draw and label points and show some coordinates, lines, and circles in Figure 4.

```
Module[
 {O, A, B, C, D, P, Q, H, k, R, X},
 {O, A, B, C, D, P, Q, H, k, R, X} =
  {{0, 0}, {0.666, 0}, {1.333, 0}, {1.666, 0}, {2, 0},
   {0.888, 0.628}, {1.777, 0.314}, {1.97, 0.246},
   {0.401, 0.801}, {1.333, 0.943}, {4, 2.8}};
 (* K is a System` symbol, so use use k instead. *)
 Graphics[{
   arbelos[0.666, 0.333],
   {Opacity[0.5], Polygon[{B, Q, R, P}]},
   Line[{{B, R}, {O, R, D}, {H, k}, {A, P}, {C, Q}}],
   Circle[R, 0.943],
   Circle[{1.333, 0.471}, 0.471],
   Circle[{1.111, 0.628}, 0.223],
   Circle[{1.555, 0.314}, 0.223],
   Circle[{1.258, 0.732}, 0.223],
   drawpoints[{O, A, B, C, D, P, Q, H, k, R}, 0.02],
   Style[{
     Text["P", P, {3.5, .6}],
     Text["Q", Q, {-1, 3}],
     Text["H", H, {-3.5, 0}],
     Text["K", k, {3.5, 0}],
     Text["R", R, {0, -2}],
     Text["O", O, {0, 2}],
     Text["A", A, {-1, 2}],
     Text["B", B, {-1, 2}],
     Text["C", C, {-1, 2}],
     Text["D", D, {-1, 2}],

     Text["(0, 0)", O, {0, 4}],
     Text[Row[{"(", Style["a", Italic], ", 0)"}], A,
      {-.5, 4}],
     Text[Row[{"2(", Style["a", Italic], ", 0)"}], B,
      {0, 4}],
     Text[Row[{"(2", Style["a", Italic], "+",
        Style["b", Italic], ", 0) "}], C, {0, 4}],
     Text[Row[{" (2(", Style["a", Italic], "+",
        Style["b", Italic], "), 0)"}], D, {0, 4}],
     Text[Row[{"(", 2, Style["a", Italic], ",2 √a b )"}],
```

```
      X, {-1.3, -1}]]},
    12
   ]
  }, PlotRange → {{-.1, 2.2}, {-.2, 1.2}},
  ImageSize → {550, 225}]]
```



▲ **Figure 4.** Labels, coordinates, lines, and circles referred to in properties 3 through 11 and 25, 26.

Property 3

The lines OR *and* DR *are orthogonal and intersect the side arcs at points* P *and* Q*, joining a common tangent to the side arcs*.

To verify the orthogonality of the lines OR and DR, we take the inner product of the vectors R and D – R.

```
Module[{a, b},
 Simplify[2 {a, √(a b)}.(2 {a + b, 0} - 2 {a, √(a b)}) == 0,
  a > 0 ⋀ b > 0]
]
```

```
True
```

We employ the following result to obtain the slopes at the points P and Q.

Theorem 1

The equation of the tangent to the left side arc at a point (x_1, y_1) *is*

$$y - y_1 = (a - x_1)/y_1 (x - x_1)$$

and the tangent to the right side arc at (x_2, y_2) *is*

$$y - y_2 = (2a + b - x_2)/y_2(x - x_2).$$

The function `PQ` finds the coordinates of the tangent points P and Q by solving a system of four equations, which places them on the arcs and sets their tangent slopes according to theorem 1.

Besides `PQ`, other definitions in this article for points and quantities are: `VWS`, `HK`, `U∪`, `EF`, `IJr`, and `LM`.

The function `dSq` computes the square of the distance between two given points.

```
dSq[a_, b_] := (b - a).(b - a)
```

```
PQ[a_, b_] := Module[
  {x1, x2, y1, y2},
  Partition[
   Simplify[
    Last /@ Last@Solve[
       {
        dSq[{x1, y1}, {a, 0}] == a^2,
        dSq[{x2, y2}, {2 a + b, 0}] == b^2,
        y2 - y1 == (a - x1) / y1 (x2 - x1),
        y1 - y2 == (2 a + b - x2) / y2 (x1 - x2)
       },
       {x1, y1, x2, y2}],
    a > 0 ∧ b > 0],
   2
  ]
 ]
```

```
Clear[a, b]; PQ[a, b]
```

$$\left\{\left\{\frac{2\,a^2}{a+b},\ \frac{2\sqrt{a^3\,b}}{a+b}\right\},\ \left\{\frac{2\,a\,(a+2\,b)}{a+b},\ \frac{2\sqrt{a\,b^3}}{a+b}\right\}\right\}$$

Property 4

The points P *and* Q *are on the radical circle.*

As BR is a diameter of the radical circle, we only need to verify the equality of the distances of P and Q to the center of the radical circle, namely the point (B + R) / 2.

```
Module[
 {a, b, P, Q, B, R},
 Map[
  Simplify[# /. Join[Thread[{P, Q} → PQ[a, b]],
      {B → {2 a, 0}, R → 2 {a, √(a b)}}], a > 0 ∧ b > 0] &,
  dSq[P, (B + R) / 2] == dSq[Q, (B + R) / 2]
 ]
]
```

```
True
```

Property 5

Let the line PQ *intersect the top arc at points* H *and* K. *Then* H *and* K *lie on a circle with center* R *and radius* BR.

We get the coordinates of the points H and K by solving a system of equations that places them on the top arc and on the line PQ .

```
HK[a_, b_] := Module[
  {x, y, x1, x2, y1, y2},
  Map[Last,
   Simplify[
    Solve[{dSq[{x, y}, {a + b, 0}] == (a + b)^2,
       (x2 - x) (y2 - y1) == (x2 - x1) (y2 - y)} /.
Thread[{x1, y1, x2, y2} → Flatten[PQ[a, b]]], {x, y}],
    a > 0 ∧ b > 0
   ],
   {2}
  ]
 ]
```

```
Clear[a, b]; HK[a, b]
```

$$\left\{\left\{\frac{2\,a\left(a^2+a\,b+2\,b\left(b+\sqrt{a^2+a\,b+b^2}\right)\right)}{(a+b)^2},\right.\right.$$
$$\left.\frac{2\sqrt{a\,b}\left(a^2-a\sqrt{a^2+a\,b+b^2}+b\left(b+\sqrt{a^2+a\,b+b^2}\right)\right)}{(a+b)^2}\right\},$$
$$\left\{\frac{2\,a\left(a^2+a\,b+2\,b\left(b-\sqrt{a^2+a\,b+b^2}\right)\right)}{(a+b)^2},\right.$$
$$\left.\left.\frac{2\sqrt{a\,b}\left(a^2+a\sqrt{a^2+a\,b+b^2}+b\left(b-\sqrt{a^2+a\,b+b^2}\right)\right)}{(a+b)^2}\right\}\right\}$$

This verifies property 5 by checking that the distances of H and K to R are the same as the distance from B to R.

```
Module[
 {H, k, R, B, a, b},
 Simplify[
  dSq[H, R] == dSq[k, R] == dSq[R, B] /.
   Join[Thread[{H, k} → HK[a, b]],
     {R → 2 {a, √(a b)}, B → {2 a, 0}}],
  a > 0 ⋀ b > 0
 ]
]
```

```
True
```

Property 6

The line AP *is parallel to the line* CQ.

This is equivalent to the fact that the determinant (cross product) of the vectors P – A and Q – C is zero.

```
Module[
 {P, Q, A, C, a, b},
 Simplify[
  0 ==
   Det[{P - A, Q - C} /. Join[Thread[{P, Q} → PQ[a, b]],
       {A → {a, 0}, C → {2 a + b, 0}}]],
  a > 0 ⋀ b > 0
 ]
]

True
```

Property 7

The line AP *is perpendicular to the line* HK.

This is equivalent to the fact that the inner product of the vectors P – A and H – K is zero.

```
Module[
 {P, A, H, k, a, b},
 Simplify[
  ((P - A).(H - k) /. Join[Thread[{P, Q} → PQ[a, b]],
       Thread[{H, k} → HK[a, b]], {A → {a, 0}}]) == 0,
  a > 0 ⋀ b > 0
 ]
]

True
```

Let us use the notation ⊙(C, r) for a circle with center C and radius r.

Property 8

The pairs O, P *and* D, Q *are inversive pairs in the circle* ⊙(R, BR).

The inversion of a point p in the circle ⊙(γ, ρ), $\gamma \neq p$ is defined to be the unique point u such that $|\gamma - p| \; |\gamma - u| = \rho^2$ [9]. The function `inversion` implements this idea.

```
inversion[Circle[γ_, ρ_], p_] := γ + ρ^2 (p - γ) / ((p - γ).(p - γ))
```

This verifies property 8, recalling the coordinates of D are $(2\,(a+b), 0)$.

```
Module[
 {P, Q, σ, a, b},
 σ = Circle[2 {a, √(a b)}, 2 √(a b)];
 Simplify[
  inversion[σ, {0, 0}] == P ∧ inversion[σ, {2 (a + b), 0}] == Q /.
   Thread[{P, Q} → PQ[a, b]],
  a > 0 ∧ b > 0
 ]
]
```

```
True
```

Property 9

Let ⊙(R, BR) *be the circle of inversion. The points* B, H, K *invert to themselves. The segment* PK *inverts to the arc* OK *and the segment* QH *inverts to the arc* DH. *The arcs* OB *and* BD *invert to themselves. The radical circle inverts to the line* OD.

Property 10

The lines AP *and* CQ *are tangent to the radical circle.*

This is the same as claiming that the corresponding arcs are orthogonal to the radical circle. By property 8, the arcs are orthogonal to the circle with diameter BR as they pass through inverse pairs [10, 11].

Property 11

BQRP *is a rectangle.*

This is one of Bankoff's surprises [12, 13, 14]. As all four points are on the radical circle, we need to verify only that PQ bisects RB.

```
Module[{a, b, P, Q},
 Simplify[{2 a, √(a b)} == (P + Q) / 2 /. Thread[{P, Q} → PQ[a, b]],
  a > 0 ∧ b > 0]
]
```

```
True
```

The following `Manipulate` illustrates properties 3–11. The easiest way to define the points `P`, `Q`, `H`, `K` is to copy and paste the formulas for them.

```
Manipulate[
 Module[{b, i, R, P, Q, H, k},
  b = 1 - a;
  i = Sqrt[a b];
  R = 2 {a, i};
  {P, Q} = {{(2 a^2)/(a + b), (2 Sqrt[a^3 b])/(a + b)}, {(2 a (a + 2 b))/(a + b), (2 Sqrt[a b^3])/(a + b)}};
  {H, k} = {{(2 a (a^2 + a b + 2 b (b + Sqrt[a^2 + a b + b^2])))/(a + b)^2, 1/(a + b)^2
      2 Sqrt[a b] (a^2 - a Sqrt[a^2 + a b + b^2] + b (b + Sqrt[a^2 + a b + b^2]))},
    {(2 a (a^2 + a b + 2 b (b - Sqrt[a^2 + a b + b^2])))/(a + b)^2, 1/(a + b)^2
      2 Sqrt[a b] (a^2 + a Sqrt[a^2 + a b + b^2] + b (b - Sqrt[a^2 + a b + b^2]))}};
  Graphics[
   {
    arbelos[a, b],
    Line[{{2 a, 0}, 2 {a, i}}],
    Line[{{0, 0}, R, {2 (a + b), 0}}],
    Line[{H, k}], Line[{{a, 0}, P}],
    Line[{{2 a + b, 0}, Q}],
    Circle[R, 2 i], Circle[{2 a, i}, i],
    {ColorData[2, 7], Opacity[.5], EdgeForm[Black],
     Polygon[{{2 a, 0}, Q, R, P}]},
    drawpoints[
     Join[{P, Q, R, {0, 0}, {a, 0}, 2 {a, 0}, {2 a + b, 0},
       2 {a + b, 0}}, {H, k}], .02]
   },
   PlotRange → {{-.1, 2.2}, {-.1, 1.2}},
   ImageSize → {425, 210}
  ]
 ],
 {{a, .6, Row[{"radius ", Style["a", Italic],
     " of left side arc"}]}, 0, 1, .001,
  Appearance → "Labeled"},
```

```
Initialization :> (
  arbelos[a_, b_] :=
   {
    Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
    White, Disk[{a, 0}, a, {0, π}],
    Disk[{2 a + b, 0}, b, {0, π}],
    Black, Circle[{a + b, 0}, a + b, {0, π}],
    Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
   };

  drawpoints[points_, s_: .05] :=
   {EdgeForm[Black], Red, Disk[#, s] & /@ points};

 )
```



■ The Incircle

Now consider the circle tangent to the side arcs and the top arc, the *incircle* $\odot(U, \upsilon)$ with tangent points V, W, and S as shown in Figure 5 [15, 16]. We also consider points X and Y at the tops of the side arcs.

```
Module[
 {O, A, B, C, D, E, F, G, U, V, W, X, Y, S, Σ},
 {O, A, B, C, D, E, F, G, U, V, W, X, Y, S, Σ} =
  {{0, 0}, {2, 0}, {4, 0}, {3, 0}, {6, 0}, {5, 0},
   {5.6, 0.8}, {3.2, 1.6}, {4.29, 1.72}, {3.6, 1.2},
   {4.61, .92}, {2, 2}, {5, 1}, {4.8, 2.4}, {4.25, 0.88}};
 Graphics[{
   arbelos[2, 1],
   ColorData[2, 6], EdgeForm[Black], Disk[U, .85],
   Gray, Line[{{A, U, E}, {C, S}, {O, S, D}, {X, D}, {O, Y}}],
   drawpoints[{X, Y, O, B, E, D, C, A, U, V, W, S, G, F, Σ}],
   Style[{
     Text["O", O, {0, 2}],
     Text["B", B, {0, 2}],
     Text["D", D, {0, 2}],
     Text[Row[{"(", Style["a", Italic], ", 0)"}], A,
      {0, 2}],
     Text[Row[{"(2", Style["a", Italic], "+",
        Style["b", Italic], ", 0)"}], E, {0, 2}],
     Text[Row[{"(", Style["a", Italic], "+",
        Style["b", Italic], ", 0)"}], C, {0, 2}],
     Text["V", V, {1, 2}],
     Text["W", W, {-2, 1}],
     Text["U", U, {-3.5, 0}],
     Text["S", S, {-3.5, -1}],
     Text["X", X, {0, 2}],
     Text["Y", Y, {-1, 2}],
     Text["E", G, {3.5, 0}],
     Text["F", F, {1, 2}],
     Text["Σ", Σ, {0, -1.5}]},
    Black, 12
   ]
  }, PlotRange → {{-.3, 6.5}, {-.4, 3.1}},
  ImageSize → {425, 210}]
]
```



▲ **Figure 5.** The incircle $\odot(U, \upsilon)$ and coordinates, lines, and points referred to in properties 12 through 15.

Proposition 6 of the *Book of Lemmas* included the value of υ, the radius of the incircle. The function `Uυ` calculates the coordinates of the center U and the radius υ.

```
Uυ[a_, b_] := Module[{x, y, υ},
  {{x, y}, υ} /.
   Last@
    Simplify[Solve[{dSq[{x, y}, {a, 0}] == (a + υ)^2,
        dSq[{x, y}, {2 a + b, 0}] == (b + υ)^2,
dSq[{x, y}, {a + b, 0}] == (a + b - υ)^2}, {x, y, υ}], a > 0 ∧ b > 0]
 ]
```

```
Clear[a, b]; Uυ[a, b]
```

$$\left\{\left\{\frac{a\left(2a^2+3ab+b^2\right)}{a^2+ab+b^2}, \frac{2ab(a+b)}{a^2+ab+b^2}\right\}, \frac{ab(a+b)}{a^2+ab+b^2}\right\}$$

The coordinates of the tangent points V, W, and S are obtained as the intersections of the lines joining the centers of the three arcs of the arbelos and the incircle.

```
VWS[a_, b_] := Module[{U, υ},
  {U, υ} = Uυ[a, b];
  Simplify[
   {{a, 0} + a Normalize[U - {a, 0}],
     {2 a + b, 0} + b Normalize[U - {2 a + b, 0}],
U + υ Normalize[U - {a + b, 0}]} /. Abs[xx_]^2 → xx^2, a > 0 ∧ b > 0]
 ]
```

```
Clear[a, b]; VWS[a, b]
```

$$\left\{\left\{\frac{2\,a\,(a+b)^2}{a^2+2\,a\,b+2\,b^2},\ \frac{2\,a\,b\,(a+b)}{a^2+2\,a\,b+2\,b^2}\right\},\ \left\{\frac{2\,a\,(a+b)\,(2\,a+b)}{2\,a^2+2\,a\,b+b^2},\ \frac{2\,a\,b\,(a+b)}{2\,a^2+2\,a\,b+b^2}\right\},\ \left\{\frac{2\,a^2\,(a+b)}{a^2+b^2},\ \frac{2\,a\,b\,(a+b)}{a^2+b^2}\right\}\right\}$$

Property 12

The points X, V, *and* D *are collinear. The points* O, W, *and* Y *are collinear. The lines* DX *and* OY *intersect in a point* Σ *lying on the incircle.*

Using the criterion of the determinant to check for collinearity, we verify the first two claims.

```
Module[
 {rules, V, W, S, a, b, D, X, Y},
 {rules = Join[Thread[{V, W, S} → VWS[a, b]],
    {D → 2 {a + b, 0}, X → {a, a}, O → {0, 0}, Y → {2 a + b, b}}]};
 Simplify[Det[{V - X, D - V} /. rules] == 0 ⋀
   Det[{O - W, Y - W} /. rules] == 0, a > 0 ⋀ b > 0]
]
```

```
True
```

Let $\Sigma(x, y)$ be the point of intersection of the lines DX and OY. Confirming that its distance to U is equal to υ verifies the third claim.

```
Module[
 {x, y, a, b, rules},
 rules = Thread[{U, υ} → Uυ[a, b]];
 Simplify[
  dSq[
    First[
     {x, y} /.
      Solve[{y / x == b / (2 a + b), y / (2 (a + b) - x) == a / (a + 2 b)},
       {x, y}]], U /. rules] == (υ^2 /. rules), a > 0 ⋀ b > 0]
]
```

```
True
```

Property 13

The points O, B, W, *and* S *are on a circle with center* X. *Similarly, the points* D, B, V, *and* S *are on a circle with center* Y.

```
Module[
 {V, W, S, X, B, U, υ, a, b},
 Simplify[dSq[W, X] == dSq[S, X] == dSq[B, X] /.
   Join[Thread[{U, υ} → Uυ[a, b]],
    Thread[{V, W, S} → VWS[a, b]], {X → {a, a}, B → {2 a, 0}}],
  a > 0 ∧ b > 0]
]
```

```
True
```

```
Module[
 {V, W, S, Y, B, U, υ, a, b},
 Simplify[dSq[V, Y] == dSq[S, Y] == dSq[B, Y] /.
   Join[Thread[{U, υ} → Uυ[a, b]],
    Thread[{V, W, S} → VWS[a, b]],
    {Y → {2 a + b, b}, B → {2 a, 0}}], a > 0 ∧ b > 0]
]
```

```
True
```

The following `Manipulate` illustrates property 13 [17]. The option for showing the Bankoff circle as the incircle of the triangle joining the center of the arcs and the incircle corresponds to property 23.

```
Manipulate[
 Module[{b, U, υ, V, W, S},
  b = 1 - a;
  {U, υ} = {{a (2 a^2 + 3 a b + b^2)/(a^2 + a b + b^2), 2 a b (a + b)/(a^2 + a b + b^2)}, a b (a + b)/(a^2 + a b + b^2)};
  {V, W, S} = {{2 a (a + b)^2/(a^2 + 2 a b + 2 b^2), 2 a b (a + b)/(a^2 + 2 a b + 2 b^2)},
    {2 a (a + b) (2 a + b)/(2 a^2 + 2 a b + b^2), 2 a b (a + b)/(2 a^2 + 2 a b + b^2)},
    {2 a^2 (a + b)/(a^2 + b^2), 2 a b (a + b)/(a^2 + b^2)}};
  Graphics[{
    arbelos[a, b],
    If[in, {EdgeForm[Black], ColorData[2, 6], Disk[U, υ]},
     {}],
```

```
      Black, Circle[{a, a}, a √2],
      Circle[{1 + a, b}, b √2],
      If[bc, {Gray, Line[{{a, 0}, {2 a + b, 0}, U, {a, 0}}],
        Black, Circle[{2 a, a b}, a b]}, {}],
      drawpoints[{V, W, S}, .03]},
  PlotRange → {{-1, 3}, {-.45, 1.5}}, ImageSize → {425, 210}]],
   {{a, .6, Row[{"radius ", Style["a", Italic],
       " of left side arc"}]}, 0, 1, .001,
    Appearance → "Labeled"},
   Row[{
     Spacer[50],
     Control@{{in, False, "incircle circle"}, {True, False}},
     Spacer[50],
     Control@{{bc, False, "Bankoff circle"}, {True, False}}
    }],

   Initialization :→ (
     arbelos[a_, b_] :=
      {
       Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
       White, Disk[{a, 0}, a, {0, π}],
       Disk[{2 a + b, 0}, b, {0, π}],
       Black, Circle[{a + b, 0}, a + b, {0, π}],
       Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
      };

     drawpoints[points_, s_: .05] :=
      {EdgeForm[Black], Red, Disk[#, s] & /@points};
    )
  ]
```



Property 14

Let Δ *be the diameter of the incircle parallel to* OD *and let* Δ' *be the projection of* Δ *onto* OD. *The rectangle between the segments* Δ *and* Δ' *is a square*.

This property is illustrated in the next `Manipulate` and is readily verified here.

```
Module[
 {a, b, U, 𝓋},
 {U, 𝓋} = U𝓋[a, b];
 Simplify[Last[U] == 2 𝓋, a > 0 ⋀ b > 0]
]

True
```

Property 15

Let E *and* F *be the intersections of the lines* OS *and* DS *with the side arcs. Then* EBFS *is a square of almost the same size as the one mentioned in property 14.*

First we obtain points E and F as the intersections of their respective lines and their respective arcs, and keep the result in the variable `replaceEF`.

```
EF[a_, b_] := Module[
  {c, d, Ex, Ey, Fx, Fy},
  {c, d} = Last@VWS[a, b];
  First[
   {{Ex, Ey}, {Fx, Fy}} /. Solve[{dSq[{Ex, Ey}, {a, 0}] == a^2,
Ey / Ex == d / c, dSq[{Fx, Fy}, {2 a + b, 0}] == b^2,
      Fy / (2 (a + b) - Fx) == d / (2 (a + b) - c)}, {Ex, Ey, Fx, Fy}]
  ]
 ]
```

```
Clear[a, b]; EF[a, b]
```

$$\left\{\left\{\frac{2\,a^3}{a^2+b^2},\ \frac{2\,a^2\,b}{a^2+b^2}\right\},\ \left\{\frac{2\,\left(a^3+a^2\,b+a\,b^2\right)}{a^2+b^2},\ \frac{2\,a\,b^2}{a^2+b^2}\right\}\right\}$$

We verify property 15 by setting E to be equal to the vector obtained by rotating F – B around B by 90° and setting S to be equal to the vector obtained by translating E by F – B.

```
Module[
 {a, b, e, F, S, B},
 (* Use e instead of E, because N[E] ⟶ 2.718… *)
 {e, F} = EF[a, b];
 S = Last@VWS[a, b];
 B = {2 a, 0};
 Simplify[(e == B + {{0, -1}, {1, 0}}.(F - B)) ⋀ (S == e + F - B),
  a > 0 ⋀ b > 0]
]
```

```
True
```

Assuming $b = 1 - a$ and $0 < a < 1$, the following plot compares the sizes of the two squares.

```
Module[{e, F},
 Plot[
  Evaluate[
   {√(dSq[e, F] / 2) /. Thread[{e, F} → EF[a, b]],
     2 υ /. Thread[{U, υ} → Uυ[a, b]]} /. b → 1 - a],
  {a, 0, 1}, AxesLabel → {a, ""}, PlotStyle → {Red, Blue}]]
```



This `Manipulate` illustrates properties 14 and 15.

```
Manipulate[
 Module[{b, e, F, U, υ, S},
  b = 1 - a;
  {e, F} = {{2 a^3/(a^2 + b^2), 2 a^2 b/(a^2 + b^2)}, {2 (a^3 + a^2 b + a b^2)/(a^2 + b^2), 2 a b^2/(a^2 + b^2)}};
  {U, υ} = {{a (2 a^2 + 3 a b + b^2)/(a^2 + a b + b^2), 2 a b (a + b)/(a^2 + a b + b^2)}, a b (a + b)/(a^2 + a b + b^2)};
  S = 2 a {a, b} / (a^2 + b^2);
  Graphics[{
    arbelos[a, b],
    EdgeForm[Black], ColorData[2, 6], Disk[U, υ],
    Gray, Line[{{0, 0}, S, 2 {a + b, 0}}],
    Opacity[.5], ColorData[2, 9],
    Rectangle[U - {υ, 0}, U + υ {1, -2}], ColorData[2, 8],
    Polygon[{e, S, F, {2 a, 0}}]}, ImageSize → {425, 210}]],
 {{a, .6, Row[{"radius ", Style["a", Italic],
     " of left side arc"}]}, 0, 1, .001,
  Appearance → "Labeled"},

 Initialization :→ (
   arbelos[a_, b_] :=
    {
     Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
     White, Disk[{a, 0}, a, {0, π}],
     Disk[{2 a + b, 0}, b, {0, π}],
     Black, Circle[{a + b, 0}, a + b, {0, π}],
     Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
    };

   drawpoints[points_, s_: .05] :=
    {EdgeForm[Black], Red, Disk[#, s] & /@ points};

  )]
```



■ The Twins

Consider the two gray circles tangent to the radical axis, a side arc, and the top arc in Figure 6. They are called *the twins*, or the *Archimedean circles*. Due to the following remarkable property, they have been extensively studied. We collect many of their extraordinary occurrences in our list of properties [3, 18, 19].

```
Module[
 {a, b, i, J, r},
 {a, b, i, J, r} = {2, 1, {3.33, 2.3}, {4.66, 1.63}, .666};
 (* Use i instead of I, because I^2 ⟶ -1 *)
 Graphics[{arbelos[2, 1], ColorData[2, 6], EdgeForm[Black],
   Disk[i, r], Disk[J, r], Black, Line[2 {{a, 0}, {a, √(a b)}}],
   Gray,
Line[{{{a, 0}, i}, {{a + b, 0}, {3.42, 2.96}},
     {{a + b, 0}, {5.14, 2.1}}, {{2 a + b, 0}, J}}],
 drawpoints[{{a, 0}, {a + b, 0}, {2 a, 0}, {2 a + b, 0}, i, J}],
   Style[{Text[Row[{"(", Style["a", Italic], ", 0)"}],
      {a, 0}, {0, 2}],
     Text[Row[{"(", Style["a", Italic], "+",
        Style["b", Italic], ", 0)"}], {a + b, 0}, {0, 2}],
     Text[Row[{"(", Style["2a", Italic], ", 0)"}],
      {2 a, 0}, {0, 2}],
```

```
      Text[Row[{"(", Style["2a", Italic], "+",
          Style["b", Italic], ", 0)"}], {2 a + b, 0}, {0, 2}],
      Text["I", i, {0, 2}],
      Text["J", J, {0, 2}]},
    Black, 12
   ]},
 PlotRange → {{-.3, 6.5}, {-.4, 3.3}}, ImageSize → {425, 210}]]
```



▲ **Figure 6.** The twins.

Property 16

The two circles tangent to the radical axis, the top arc, and one of the side arcs of an arbelos have the same radius.

This property appeared as proposition 5 in the *Book of Lemmas*. Solving the following system of six equations finds the values of the radii, verifies they are equal, and computes the centers I, J.

```
Clear[a, b];
Module[{i, Ix, Iy, J, Jx, Jy, r1, r2 (* radii *)},
 i = {Ix, Iy};
 J = {Jx, Jy};
Simplify[{i, J, r1, r2} /.
   Solve[{dSq[i, {a, 0}] == (a + r1)^2,
     dSq[J, {2 a + b, 0}] == (b + r2)^2,
     dSq[i, {a + b, 0}] == (a + b - r1)^2,
dSq[J, {a + b, 0}] == (a + b - r2)^2, Ix + r1 == Jx - r2 == 2 a},
    {Ix, Iy, Jx, Jy, r1, r2}], a > 0 ∧ b > 0]]
```

$$\left\{\left\{\left\{\frac{a\,(2\,a+b)}{a+b},\,-2\,a\,\sqrt{\frac{b}{a+b}}\right\},\,\left\{\frac{a\,(2\,a+3\,b)}{a+b},\,-2\,b\,\sqrt{\frac{a}{a+b}}\right\},\right.\right.$$

$$\left.\frac{a\,b}{a+b},\,\frac{a\,b}{a+b}\right\},\,\left\{\left\{\frac{a\,(2\,a+b)}{a+b},\,-2\,a\,\sqrt{\frac{b}{a+b}}\right\},\right.$$

$$\left.\left\{\frac{a\,(2\,a+3\,b)}{a+b},\,2\,b\,\sqrt{\frac{a}{a+b}}\right\},\,\frac{a\,b}{a+b},\,\frac{a\,b}{a+b}\right\},$$

$$\left\{\left\{\frac{a\,(2\,a+b)}{a+b},\,2\,a\,\sqrt{\frac{b}{a+b}}\right\},\,\left\{\frac{a\,(2\,a+3\,b)}{a+b},\,-2\,b\,\sqrt{\frac{a}{a+b}}\right\},\right.$$

$$\left.\frac{a\,b}{a+b},\,\frac{a\,b}{a+b}\right\},\,\left\{\left\{\frac{a\,(2\,a+b)}{a+b},\,2\,a\,\sqrt{\frac{b}{a+b}}\right\},\right.$$

$$\left.\left.\left\{\frac{a\,(2\,a+3\,b)}{a+b},\,2\,b\,\sqrt{\frac{a}{a+b}}\right\},\,\frac{a\,b}{a+b},\,\frac{a\,b}{a+b}\right\}\right\}$$

These four solutions give the centers in pairs: (I*, J*), (I*, J), (I, J*), (I, J), where I* and J* are the reflections of I and J in the diameter of the arbelos; only the last expression is valid. The result also shows that the twins are indeed of the same radius $r = a\,b/(a+b)$. Any circle with radius equal to the twins' radius is called *Archimedean*. A nice interpretation of r arises when considering a and b as resistances: then r is the resistance resulting from connecting a and b in parallel; that is, $1/r = 1/a + 1/b$. The function `IJr` computes the value of the centers and the common value of the radius of the twins.

$$\mathbf{IJr[a\_,\ b\_]:=}\left\{\left\{\frac{a\,(2\,a+b)}{a+b},\,2\,a\,\sqrt{\frac{b}{a+b}}\right\},\,\left\{\frac{a\,(2\,a+3\,b)}{a+b},\,2\,b\,\sqrt{\frac{a}{a+b}}\right\},\,\frac{a\,b}{a+b}\right\}$$

Property 17

The area of the arbelos is equal to the area of the smallest circle enclosing the twins.

Consider a circle tangent to both twins, with center at point U = (x, y) and radius v. Then there are two possible values of v.

```
possibleV[x_] := Module[{i, J, r, U, y, v},
  {i, J, r} = IJr[a, b];
  U = {x, y};
  Simplify[v /. Solve[dSq[U, i] == dSq[U, J] == (v - r)^2, {y, v}],
   a > 0 ∧ b > 0]]
```

```
Clear[a, b, x];
possibleV[x]
```

$$\left\{\frac{1}{(a-b)^2 (a+b)}\left(a^3 b-2 a^2 b^2+a b^3+\sqrt{\left(1/(a+b)\ (a-b)^2\left(4 a^7+a^6 (9 b-4 x)+8 a^{9/2} b^{3/2} (2 b-x)-8 a^{5/2} b^{7/2} x+b^5 x^2+2 a^{7/2} b^{3/2} x (-8 b+x)-2 a^{3/2} b^{7/2}\left(b^2-x^2\right)+a^5\left(6 b^2+6 b \sqrt{a b}-8 b x+x^2\right)+a^3 b^2\left(5 b^2+12 b \sqrt{a b}-4 b x+x^2\right)+a^4 b\left(b^2-4 b x+2 x^2\right)+a b^4\left(b^2-4 b x+2 x^2\right)+a^2 b^2\left(6 b^3-8 b^2 x+b x^2+4 \sqrt{a b}\, x^2\right)\right)\right)}\right),\right.$$

$$\left.\frac{1}{(a-b)^2 (a+b)}\left(a^3 b-2 a^2 b^2+a b^3-\sqrt{\left(1/(a+b)\ (a-b)^2\left(4 a^7+a^6 (9 b-4 x)+8 a^{9/2} b^{3/2} (2 b-x)-8 a^{5/2} b^{7/2} x+b^5 x^2+2 a^{7/2} b^{3/2} x (-8 b+x)-2 a^{3/2} b^{7/2}\left(b^2-x^2\right)+a^5\left(6 b^2+6 b \sqrt{a b}-8 b x+x^2\right)+a^3 b^2\left(5 b^2+12 b \sqrt{a b}-4 b x+x^2\right)+a^4 b\left(b^2-4 b x+2 x^2\right)+a b^4\left(b^2-4 b x+2 x^2\right)+a^2 b^2\left(6 b^3-8 b^2 x+b x^2+4 \sqrt{a b}\, x^2\right)\right)\right)}\right)\right\}$$

To find the extrema of v, we set the derivative of each of the above expressions to zero and solve for x.

```
Clear[a, b, x];
Simplify[Map[Solve[D[#, x] == 0, x] &, possibleV[x]],
 a > 0 ⋀ b > 0]
```

```
{{{x → 2 a}}, {{x → 2 a}}}
```

So the centers of the smallest and largest circles tangent to the twins lie on the radical axis. Moreover, they are concentric, as this result confirms.

```
Clear[a, b];
Module[{i, J, r, U, y, v},
 {i, J, r} = IJr[a^2, b^2];
 U = {2 a^2, y};
 FullSimplify[
  Simplify[{U, v} /. Solve[dSq[U, i] == dSq[U, J] == (v - r)^2,
      {y, v}], a > 0 ⋀ b > 0] /. {a → Sqrt[a], b → Sqrt[b]}, a > 0 ⋀ b > 0]]
```

$$\left\{\left\{\left\{2\,a,\ \frac{a\sqrt{b}+\sqrt{a}\,b}{\sqrt{a+b}}\right\},\ -\sqrt{a\,b}+\frac{2\,a\,b}{a+b}\right\},\ \left\{\left\{2\,a,\ \frac{a\sqrt{b}+\sqrt{a}\,b}{\sqrt{a+b}}\right\},\ \sqrt{a\,b}\right\}\right\}$$

Thus, by using property 2, we confirm that the largest tangent circle, which is the smallest enclosing the twins, satisfies property 17. The following `Manipulate` shows the circles tangent to the twins as you vary the radius *a* of the left side arc.

```
Manipulate[
 Module[{b, h, k, r, r2, u},
  b = 1 - a;
  h = {a (2 a + b), 2 a √b};
  k = {a (2 a + 3 b), 2 √a b};
  r = a b;
  r2 = Norm[h - k] / 2 - r;
  u = {2 a, a √b + √a b};
  Graphics[
   {arbelos[a, b], EdgeForm[Black], ColorData[2, 6],
    Disk[h, r], Disk[k, r], Black,
    Line[{{2 a, 0}, 2 {a, √(a b)}}], Circle[u, √(a b)],
    Circle[u, r2]},
   PlotRange → {{-.3, 2.3}, {-.1, 1.25}},
   ImageSize → {425, 210}]
 ],

 {{a, .7, Row[{"radius ", Style["a", Italic],
     " of left side arc"}]}, 0, 1, .001,
  Appearance → "Labeled"},

 Initialization :→ (
   arbelos[a_, b_] :=
    {
     Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
     White, Disk[{a, 0}, a, {0, π}],
     Disk[{2 a + b, 0}, b, {0, π}],
     Black, Circle[{a + b, 0}, a + b, {0, π}],
     Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
    }
  )
]
```



The following plot compares the radii of the two circles tangent to the twins with centers on the radical axis.

```
Module[
 {a, i, J, r},
 {i, J, r} = IJr[a, 1 - a];
 Plot[{Norm[i - J] / 2 - r, √r}, {a, 0, 1}]
]
```



```
Module[
 {a, b, r, i, J, L, M, Q, Z, G},
 {a, b, r, i, J, L, M, Q, Z, G} =
  {2, 1, 0.66, {3.33, 2.3}, {4.66, 1.63}, {3, 1.73},
   {4.8, 0.97}, {4, 2.82}, {4, 2.3}, {4, 1.63}};
 Graphics[{
   arbelos[a, b],
   EdgeForm[Black], ColorData[2, 6], Disk[i, r],
   Disk[J, r], Disk[{2 a - r, 0}, r], Disk[{2 a + r, 0}, r],
   Black, Line[{2 {a, 0}, Q}],
   {
    Gray,
    Line[{{L, 2 {a + b, 0}}, {{0, 0}, M},
      2 {{a + b, 0}, {a, √(a b)}, {0, 0}}, {i, Z}, {J, G}}]
   },
   drawpoints[{{0, 0}, {2 a, 0}, 2 {a + b, 0}, Q, L, M,
     i, J, G, Z}], Style[{Text["O", {0, 0}, {0, 2}],
     Text["D", 2 {a + b, 0}, {0, 2}],
     Text["R", 2 {a, √(a b)}, {-3, -1}],
     Text["Z", Z, {-3.5, 0}],
     Text["G", G, {-3.5, 0}],
     Text["B", {2 a, 0}, {2, 2}],
     Text["L", L, {0, 2}],
     Text["M", M, {0, 1.5}],
     Text["I", i, {0, 2}],
     Text["J", J, {0, 2}]},
    12
   ]},
PlotRange → {{-.1, 2 (a + b) + .2}, {-.7, a + b + .1}},
  ImageSize → {425, 210}]]
```



▲ **Figure 7.** Labels and lines referred to in properties 18 through 24.

Property 18

The common tangent of the left arc and its tangent twin at L *passes through* D. *Similarly, the common tangent of the right arc and its tangent twin at* M *passes through* O (see Figure 7).

This computes the tangent points L and M.

```
LM[a_, b_] := Module[{i, J, r},
  {i, J, r} = IJr[a, b];
  Simplify[{{a, 0} + a Normalize[i - {a, 0}],
    {2 a + b, 0} + b Normalize[J - {2 a + b, 0}]}, a > 0 ∧ b > 0]]
```

```
Clear[a, b]; LM[a, b]
```

$$\left\{\left\{\frac{2\,a\,(a+b)}{a+2\,b},\ \frac{2\,a\sqrt{b\,(a+b)}}{a+2\,b}\right\},\ \left\{\frac{4\,a\,(a+b)}{2\,a+b},\ \frac{2\,b\sqrt{a\,(a+b)}}{2\,a+b}\right\}\right\}$$

By using theorem 1, we verify both claims.

```
Module[{x1, y1, x2, y2, Lx, Ly, Mx, My},
 {x1, y1} = {2 (a + b), 0};
 {x2, y2} = {0, 0};
 {{Lx, Ly}, {Mx, My}} = LM[a, b];
 Simplify[(y1 - Ly == (a - Lx) / Ly (x1 - Lx)) ∧
   (y2 - My == (2 a + b - Mx) / My (x2 - Mx))]
]
```

```
True
```

Property 19

The length DL *is equal to the length* DR. *The length* OM *is equal to the length* OR.

We verify both claims simultaneously.

```
Module[{L, M, D, R, O},
 {L, M} = LM[a, b];
 {R, D, O} = {2 {a, √(a b)}, {2 (a + b), 0}, {0, 0}};
 Simplify[(dSq[D, L] == dSq[D, R]) ∧ (dSq[O, M] == dSq[O, R])]]
```

```
True
```

However, the points L, I, and R are not on a circle centered at D, nor are the points R, J, and M on a circle centered at O; otherwise, the following expression would be zero.

```
Clear[a, b];
Simplify[(J.J /. (J → IJr[a, b][[2]])) -
   (M.M /. (M → Last[LM[a, b]]))]
```

$$\frac{a^2 b^2}{(a+b)^2}$$

Property 20

The line DL *bisects the segment* BZ. *The line* OM *bisects the segment* BG.

As the length of the segment BZ is the ordinate of I and the length of the segment BG is the ordinate of J, we only need to verify that the midpoints of those segments lie on the mentioned lines by checking slopes.

```
Module[{L, M, i, J},
 {L, M} = LM[a, b];
 {i, J} = Take[IJr[a, b], 2];
 Simplify[
   (Last[i] / 2 / (2 b) == Last[L] / (2 (a + b) - First[L])) ∧
    (Last[J] / 2 / (2 a) == Last[M] / First[M]), a > 0 ∧ b > 0]]
```

```
True
```

Property 21

The two blue circles with diameters on OD *passing through* B *tangent to the lines* OM *and* DL *are Archimedean.*

Those circles are the fourth and fifth Archimedean circles discovered by Bankoff [20]. In order to verify this property, we use the following result [21]:

Theorem 2

The distance of the point $P(x, y)$ *to the line passing through different points* $A(x_1, y_1)$ *and* $B(x_2, y_2)$ *is*

$$\frac{\begin{vmatrix} x_2 - x_1 & y_2 - y_1 \\ x_1 - x & y_1 - y \end{vmatrix}}{\sqrt{(x_2 - x_1)^2 + (y_2 - y_1)^2}}.$$

This directed distance is positive if the triangle PAB is traversed counterclockwise and negative otherwise. The function `dAB` implements this.

```
dAB[P_, A_, B_] := Det[{B - A, A - P}] / Sqrt[dSq[A, B]]
```

Let $(x, 0)$ and r_1 be the center and radius of the blue circle on the left side of point B in Figure 7. Solving the following system finds the value of r_1.

```
Clear[a, b];
Simplify[
 r /. Solve[{dAB[{x, 0}, {0, 0}, Last[LM[a, b]]] == r,
    x + r == 2 a}, {x, r}]]
```

$$\left\{\frac{a\,b}{a+b}\right\}$$

Similarly, this calculates the radius of the blue circle to the right of B, which equals r_1.

```
Clear[a, b];
Module[{x, r},
 Simplify[
  r /.
   Solve[
    {dAB[{x, 0}, First[LM[a, b]], {2 (a + b), 0}
       (* point D *)] == r, x - r == 2 a}, {x, r}]]
]
```

$$\left\{\frac{a\,b}{a+b}\right\}$$

Thus, both circles are Archimedean as claimed. The following `Manipulate` shows the twins and these two other circles.

```
Manipulate[
 Module[{b, i, J, r, t1, t2, R, z, g},
  b = 1 - a;
  i = a {1 + a, 2 √b};
  J = {a (2 + b), 2 √a b};
  r = a b;
  t1 = 2 a {1, √b} / (1 + b);
  t2 = 2 {2 a , b √a} / (1 + a);
  R = {a, √(a b)};
  z = 2 a {1, √b};
  g = 2 {a, √a b};
  Graphics[{arbelos[a, b], EdgeForm[Black],
     ColorData[2, 6], Disk[i, r], Disk[J, r],
     Disk[{2 a - r, 0}, r], Disk[{2 a + r, 0}, r], Black,
     Line[{2 {a, 0}, 2 R}], Gray, Line[{t1, 2 {a + b, 0}}],
     Line[{{0, 0}, t2}], Line[2 {{a + b, 0}, R, {0, 0}}]},
   PlotRange → {{-.1, 2 (a + b) + .2}, {-.28, a + b + .1}},
   ImageSize → {425, 210}]],
 {{a, .6, Row[{"radius ", Style["a", Italic],
      " of left side arc"}]}, 0, 1, .001,
  Appearance → "Labeled"},

 Initialization :→ (
   arbelos[a_, b_] :=
    {
     Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
     White, Disk[{a, 0}, a, {0, π}],
     Disk[{2 a + b, 0}, b, {0, π}],
     Black, Circle[{a + b, 0}, a + b, {0, π}],
     Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
    }
  )
]
```



Property 22

The circle through V, W, *and* B *in Figure 5, called the Bankoff circle, is Archimedean*.

Archimedes discovered the original twins; Bankoff improved on this by discovering this third circle in 1950 [22]. The coordinates of the center T of the Bankoff circle are obtained by equating the distances of T to the points V, W, and B.

```
Clear[a, b]; Module[{v, w, x, y},
 {v, w} = Take[VWS[a, b], 2];
 Last /@
  First@Solve[dSq[{x, y}, v] == dSq[{x, y}, w] ==
     dSq[{x, y}, {2 a, 0} (* point B *)], {x, y}]
]
```

$\left\{2\,a,\ \frac{a\,b}{a+b}\right\}$

Property 23

The Bankoff circle is the incircle of the triangle formed by joining the centers of the side arcs and the center U *of the incircle of the arbelos.*

Using theorem 2 to compute the distance of U to the sides of the triangle, we verify this property (as dAB computes a directed distance, the order of the arguments describing the line is important).

```
Module[{a, b, bc, U, A, C},
 U = First@U∪[a, b];
 bc = {2 a, a b/(a + b)};
 {A, C} = {{a, 0}, {2 a + b, 0}};
 Simplify[dAB[bc, A, U] == dAB[bc, U, C], a > 0 ∧ b > 0]]
```

```
True
```

Property 24

The circle $A_4 = \odot(\alpha, \gamma)$ *tangent to the circles* $\odot(\mathrm{O}, 2\,a)$, $\odot(\mathrm{D}, 2\,b)$, *and the top arc is Archimedean.*

This computes the values of α and γ.

```
Clear[a, b]; Module[{α, x, y, γ},
 α = {x, y};
 Simplify[
  {α, γ} /.
   Solve[{dSq[α, {0, 0}] == (2 a + γ)^2,
     dSq[α, {2 (a + b), 0}] == (2 b + γ)^2,
     dSq[α, {a + b, 0}] == (a + b - γ)^2}, {x, y, γ}],
  a > 0 ∧ b > 0
 ]
]
```

$$\left\{\left\{\left\{\frac{a\left(2a^2+5ab+b^2\right)}{(a+b)^2}, -\frac{2ab\sqrt{2a^2+5ab+2b^2}}{(a+b)^2}\right\}, \frac{ab}{a+b}\right\}, \left\{\left\{\frac{a\left(2a^2+5ab+b^2\right)}{(a+b)^2}, \frac{2ab\sqrt{2a^2+5ab+2b^2}}{(a+b)^2}\right\}, \frac{ab}{a+b}\right\}\right\}$$

The circle A_4 is the one where the ordinate of α is positive. Note that α is not on the radical axis.

```
Manipulate[
 Module[{b},
  b = 1 - a;
  Graphics[{arbelos[a, b], EdgeForm[Black],
    ColorData[2, 6],
    Disk[{a (2 a^2 + 5 a b + b^2), 2 a b Sqrt[(2 a + b) (a + 2 b)]}, a b],
    Black, Circle[{2, 0}, 2 b], Circle[{0, 0}, 2 a],
    Line[2 {{a (a - b), 2 a^(3/2) Sqrt[b]}, {a (a + 3 b), 2 Sqrt[a] b^(3/2)}}]},
   PlotRange → {{-.3, 2.4}, {-.1, 1.3}},
   ImageSize → {425, 210}]],
 {{a, .6, Row[{"radius ", Style["a", Italic],
     " of left side arc"}]}, 0, 1, .001,
  Appearance → "Labeled"},

 Initialization :→ (
   arbelos[a_, b_] :=
    {
     Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
     White, Disk[{a, 0}, a, {0, π}],
     Disk[{2 a + b, 0}, b, {0, π}],
     Black, Circle[{a + b, 0}, a + b, {0, π}],
     Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
    }
  )]
```



Property 25

The circles A_5 *and* A_6 *tangent to the radical axis, one passing through* P *and the other passing through the point* Q*, are both Archimedean* (see Figure 4).

```
Clear[a, b]; Module[{P, Q},
 {P, Q} = PQ[a, b];
 Simplify[{2 a - First[P], First[Q] - 2 a} / 2]]
```

$$\left\{\frac{a\,b}{a+b}, \frac{a\,b}{a+b}\right\}$$

Property 26

The circle A_7 *tangent to the line* HK *and the top arc at* R *is Archimedean* (see Figure 4).

A circle with center S(x, y) and radius r tangent to the line HK is such that the distance from S to HK is r, so this equation holds:

$$\left| S - K - (S-K)\cdot(H-K)\,\frac{(H-K)}{|\,H-K\,|^2} \right| = r.$$

Because the circle passes through R,

$$|\,S - R\,| = r.$$

Because the circle A_7 is tangent to the top arc,

$$|\,(a+b, 0) - S\,| = a + b - r.$$

This input uses explicit expressions for x, y, and r that satisfy these three equations.

```
Module[{r, S, H, k},
 r = a b / (a + b);
 S = {a (2 a^2 + 3 a b + 3 b^2) / (a + b)^2, 2 Sqrt[a b] (a^2 + a b + b^2) / (a + b)^2};
 {H, k} = HK[a, b];
 Simplify[{dAB[S, H, k] == r,
   dSq[2 {a, Sqrt[a b]} (* point R *), S] == r^2,
   dSq[{a + b, 0}, S] == (a + b - r)^2}, a > 0 ∧ b > 0]]
```

```
{True, True, True}
```

Property 27

Consider the two (red) segments connecting the center of the top arc to the top points X *and* Y *of the left and right arcs of the arbelos. These segments have the same length and are orthogonal. The tangent circles* A_8 *and* A_9 *at* X *and* Y *to those lines and the top arc are Archimedean* (see Figure 8).

This property was discovered in the summer of 1998 [23].

```
Module[{a = 2, b = 1, r, a1, a2, a3, a4},
 r = a b / (a + b);
 a1 = {a, a} + r Normalize[{a, b}];
 a2 = {a, a} - r Normalize[{a, b}];
 a3 = {2 a + b, b} + r Normalize[{-b, a}];
 a4 = {2 a + b, b} - r Normalize[{-b, a}];
 Graphics[{
   arbelos[a, b],
   {Opacity[.8], EdgeForm[Black], ColorData[2, 6],
    Disk[a1, r], Disk[a2, r], Disk[a3, r], Disk[a4, r]},
   Red, Line[{{a, a}, {a + b, 0}, {2 a + b, b}}],
   drawpoints[{{a + b, 0}, {a, a}, {2 a + b, b}}],
   Style[{
     Text[Row[{"(", Style["a", Italic], "+",
        Style["b", Italic], ", 0)"}], {a + b, 0}, {0, 2}],
     Text["Y", {2 a + b, b}, {0, 2}],
     Text["X", {a, a}, {2, 2}]},
    12, Black]
  },
  PlotRange → {{-.3, 6.5}, {-.4, 3.3}}, ImageSize → {425, 210}
 ]
]
```



▲ **Figure 8.** The two pairs of Archimedean circles from property 27.

Slanted Twins

We have seen that there are some Archimedean circles other than the twins, namely the Bankoff circle and those mentioned in properties 21 through 27. There are also *non-Archimedean twins*, that is, pairs of circles of the same radius, different than that of the twins, appearing at significant places within the arbelos.

The discovery of the *slanted twins* arose from the initial assumption that, besides being tangent to either side arc and the top arc, the two circles-to-be-twins could be tangent to themselves and not necessarily to the radical axis. Clearly there are an infinite number of solutions if we do not require these circles to be of equal radius. The idea was that if we started by assuming they are of equal radius, we might end up discovering they are tangent to the radical axis. This turned out not to be the case. Let us consider circles with centers at the points $h(x_1, y_1)$ and $k(x_2, y_2)$ with common radius r. The value of r can be obtained by solving a system of five equations.

```
Clear[a, b]; Module[
 {x1, y1, x2, y2, r},
 Union@Simplify[r /. Solve[
     {
      dSq[{x1, y1}, {a, 0}] == (a + r)^2,
      dSq[{x1, y1}, {a + b, 0}] == (a + b - r)^2,
      dSq[{x2, y2}, {2 a + b, 0}] == (b + r)^2,
      dSq[{x2, y2}, {a + b, 0}] == (a + b - r)^2,
      dSq[{x1, y1}, {x2, y2}] == 4 r^2, r ≠ a, r ≠ b
     }, {x1, y1, x2, y2, r}], a > 0 ∧ b > 0]
]
```

$$\left\{\frac{(a+b)\left(a^2+2\,a\,b+b^2-\sqrt{a^4+14\,a^2\,b^2+b^4}\right)}{2\,(a-b)^2},\ \frac{(a+b)\left(a^2+2\,a\,b+b^2+\sqrt{a^4+14\,a^2\,b^2+b^4}\right)}{2\,(a-b)^2}\right\}$$

These expressions involve square roots differing in sign. The ones using the plus sign diverge at $a = b$ and are rejected.

```
Module[{a, b},
 Limit[((a + b) (a^2 + 2 a b + b^2 + Sqrt[a^4 + 14 a^2 b^2 + b^4]))/(2 (a - b)^2) /. b → 1 - a,
  a → 1 / 2]]
```

∞

The other one converges.

```
Module[{a, b},
 Limit[((a + b) (a^2 + 2 a b + b^2 - Sqrt[a^4 + 14 a^2 b^2 + b^4]))/(2 (a - b)^2) /. b → 1 - a,
  a → 1 / 2]]
```

$\frac{1}{4}$

We conclude that the slanted twins are indeed congruent and that their common radius r is

$$\frac{(a+b)\left(a^2+2\,a\,b+b^2-\sqrt{a^4+14\,a^2\,b^2+b^4}\right)}{2\,(a-b)^2}.$$

The following comparison between the radii of the twins and the slanted twins shows that their difference turns out to be very small.

```
Module[{a, b},
 Plot[
  Evaluate[{a b/(a + b), ((a + b) (a^2 + 2 a b + b^2 - Sqrt[a^4 + 14 a^2 b^2 + b^4]))/(2 (a - b)^2)} /.
    b -> 1 - a], {a, 0, 1}]]
```



This gives the coordinates of the centers of the slanted twins.

```
centers[a_, b_] := Module[
  {r, x1, y1, x2, y2},
  r = ((a + b) (a^2 + 2 a b + b^2 - Sqrt[a^4 + 14 a^2 b^2 + b^4]))/(2 (a - b)^2);
  First@FullSimplify[
    {{x1, y1}, {x2, y2}} /.
     Solve[{dSq[{x1, y1}, {a, 0}] == (a + r)^2, dSq[{x1, y1},
{a + b, 0}] == (a + b - r)^2, dSq[{x2, y2}, {2 a + b, 0}] == (b + r)^2,
       dSq[{x2, y2}, {a + b, 0}] == (a + b - r)^2,
dSq[{x1, y1}, {x2, y2}] == 4 r^2}, {x1, y1, x2, y2}],
    a > 0 ∧ b > 0
   ]
 ]
```

```
Clear[a, b]; centers[a, b]
```

$$\left\{\left\{\frac{(a+b)\,(2\,a+b)\,\left((a+b)^2-\sqrt{a^4+14\,a^2\,b^2+b^4}\right)}{2\,(a-b)^2\,b},\right.\right.$$

$$\frac{1}{(a-b)^2\,b}\sqrt{2}\,a\,(a+b)\,\sqrt{\left(-a^4-2\,a^3\,b-3\,b^4+5\,b^2\sqrt{a^4+14\,a^2\,b^2+b^4}+a^2\left(-12\,b^2+\sqrt{a^4+14\,a^2\,b^2+b^4}\right)+2\,a\,b\left(-7\,b^2+\sqrt{a^4+14\,a^2\,b^2+b^4}\right)\right)}\Bigg\},$$

$$\left\{\frac{1}{2\,a\,(a-b)^2}\,(a+b)\left(3\,a^3-12\,a^2\,b-a\,b^2-2\,b^3+a\sqrt{a^4+14\,a^2\,b^2+b^4}+2\,b\sqrt{a^4+14\,a^2\,b^2+b^4}\right),\right.$$

$$\frac{1}{2\sqrt{2}\,a\,(a-b)^2\,b}\,(a+b)\left(a^2-b^2+\sqrt{a^4+14\,a^2\,b^2+b^4}\right)\sqrt{\left(-a^4-2\,a^3\,b-3\,b^4+5\,b^2\sqrt{a^4+14\,a^2\,b^2+b^4}+a^2\left(-12\,b^2+\sqrt{a^4+14\,a^2\,b^2+b^4}\right)+2\,a\,b\left(-7\,b^2+\sqrt{a^4+14\,a^2\,b^2+b^4}\right)\right)}\Bigg\}\Bigg\}$$

The following `Manipulate` shows the slanted twins and, optionally, the twins, as you vary a.

```
Manipulate[
 Module[{b, c, d, sr, s1, s2},
  b = 1 - a;
  c = Sqrt[a^4 + 14 a^2 b^2 + b^4];
  d =
   Sqrt[(-a^4 - 2 a^3 b - 3 b^4 + 5 b^2 c + a^2 (-12 b^2 + c) + 2 a b (-7 b^2 + c))];
  sr = If[a == 1 / 2, 1 / 4, ((a + b) ((a + b)^2 - c))/(2 (a - b)^2)];
  s1 = If[
    a == 1 / 2,
    {3/4, 1/Sqrt[2]},
    {((a + b) (2 a + b) ((a + b)^2 - c))/(2 (a - b)^2 b), 1/((a - b)^2 b) Sqrt[2] a (a + b) d}
   ];
  s2 = If[
    a == 1 / 2,
    {5/4, 1/Sqrt[2]},
    {((a + b) (3 a^3 - 12 a^2 b - a b^2 - 2 b^3 + a c + 2 b c))/(2 a (a - b)^2),
     1/(2 Sqrt[2] a (a - b)^2 b) (a + b) (a^2 - b^2 + c) d}
   ];
  Graphics[{arbelos[a, b], EdgeForm[Black], ColorData[2, 6],
    Disk[s1, sr], Disk[s2, sr], Black,
    Line[2 {{a, 0}, {a, Sqrt[a b]}}],
    If[st, {Red, Circle[a {1 + a, 2 Sqrt[b]}, a b],
      Circle[{a (2 + b), 2 Sqrt[a] b}, a b]}]},
   PlotRange -> {{-.1, 2.2}, {-.15, 1.2}},
   ImageSize -> {425, 210}]],
```

- -

```
{{a, .3, Row[{"radius ", Style["a", Italic],
    " of left side arc"}]}, 0, 1, .001,
 Appearance → "Labeled"},
{{st, False, "show twins"}, {True, False}},

Initialization :→ (
  arbelos[a_, b_] :=
   {
    Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
    White, Disk[{a, 0}, a, {0, π}],
    Disk[{2 a + b, 0}, b, {0, π}],
    Black, Circle[{a + b, 0}, a + b, {0, π}],
    Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
   }

 )]
```



Generalizations

In this section we generalize the shape of an arbelos by allowing the arcs to cross and by considering a 3D version. To set the context of the first of those generalizations, we need the concept of the *radical axis of two circles*.

Radical Axis

Let p be a point and σ be the circle $\odot(h, r)$. The *power* of p with respect to σ is defined to be the real number $\left|h - p\right|^2 - r^2$. The power of p is positive, zero, or negative depending on whether p lies outside, on, or inside σ [12]. Let $f(x, y) = a\left(x^2 + y^2\right) + b\,x + c\,y + d$; if the points of σ satisfy the equation $f(x, y) = 0$, then an alternative way to define the power of $p = (u, v)$ is to evaluate $f(u, v)$. (A similar result applies if $a = 0$, when the circle degenerates to a line, in which case the sign of $f(u, v)$ indicates whether p is above, on, or below the line.)

Here is a very interesting property of the power of a point. Given a circle and a point p, choose an arbitrary line through p meeting the circle at points q and r. Then the product $|p - q|\,|q - r|$ depends only on p—it is independent of the choice of line through p. This product is equal to the power of p.

In the following `Manipulate`, drag the four locators to vary the size of the circle, the position of *p*, and the slope of the line through *p*.

```
Manipulate[
 Module[{c, a, p, q, pw, r, α, s, x, y},
  {c, a, p, q} = capq;
  pw = Norm[p - c]^2 - r^2;
  r = Norm[a - c];
  s = NSolve[dSq[α p + (1 - α) q, c] == r^2, α, Reals];
  If[s ≠ {}, {x, y} = α p + (1 - α) q /. s];
  Graphics[{
    EdgeForm[Black], LightOrange, Disk[c, r], Black,
    Line[{p + 20 Normalize[p - q], p - 20 Normalize[p - q]}],
    Red, If[s ≠ {}, Disk[#, .1] & /@ {x, y}],
    Style[{
      Text[Style["p", Italic], p, {0, 2}],
      Text[Row[{"power of ", Style["p", Italic],
          " = ", NumberForm[pw, {6, 4}], " ⇒ ",
          Style["p", Italic],
          Which[pw < 0, " is inside", pw == 0, " is on",
           True, " is outside"], " the circle"}],
       {0, -3.5}, {0, 0}],
      If[s ≠ {}, {Text[Style["q", Italic], x, {0, 2}],
        Text[Style["r", Italic], y, {0, 2}],
        Text[Row[{"| ", Style["p", Italic], "-",
           Style["q", Italic], " | | ", Style["q", Italic],
           "-", Style["r", Italic], " | = ",
           NumberForm[Norm[p - x] Norm[p - y], {6, 4}]}],
         {0, -3.85}, {0, 0}]}, {}]}, Black, 12]},
   PlotRange → 4, ImageSize → {400, 400}
  ]
 ],

 {{capq, {{1, 0}, {-1, -.5}, {-1, 2}, {2, 1}}}, {-4, -4},
  {4, 4}, Locator},

 Initialization :→ (
   arbelos[a_, b_] :=
    {
     Lighter@ColorData[5, 7], Disk[{a + b, 0}, a + b, {0, π}],
     White, Disk[{a, 0}, a, {0, π}],
     Disk[{2 a + b, 0}, b, {0, π}],
     Black, Circle[{a + b, 0}, a + b, {0, π}],
     Circle[{a, 0}, a, {0, π}], Circle[{2 a + b, 0}, b, {0, π}]
    };
   dSq[a_, b_] := (b - a).(b - a)
  )]
```



Given two circles with different centers, their *radical axis* is defined to be the line consisting of all points that have equal powers with respect to each of the two circles. Proofs of the following can be found in [10].

Theorem 3

If two circles intersect at two points B_1 *and* B_2*, then their radical axis is the common secant* $\mathrm{B}_1\,\mathrm{B}_2$*. If two circles are tangent at* B*, then their radical axis is their common tangent at* B.

Corollary 1

Given three circles with noncollinear centers, the three radical axes of the circles taken in pairs are distinct concurrent lines.

Theorem 4

The radical axis of two circles is the locus of points from which tangents drawn to both circles have the same length.

The following `Manipulate` shows two circles; one is fixed, and you can vary the center and size of the other one by dragging the locator or changing its radius with the slider. You can use the other slider to move the red point on the radical axis to illustrate theorem 4.

```
Manipulate[
 Module[{α, d, p1, p2, p, t1, t2},
  If[Chop[Norm[cen]] == 0, cen = {.01, .01}];
  α = (Norm[cen]^2 - r^2 + 1) / (2 Norm[cen]^2);
  d = Normalize[{-Last[cen], First[cen]}];
  {p1, p2} = {α cen + 4 d, α cen - 4 d};
  p = rp p1 + (1 - rp) p2;
  If[
   Norm[p] > 1,
   t1 = Normalize[RotationMatrix[ArcCos[1 / Norm[p]]].p];
   t2 =
    cen +
     r Normalize[RotationMatrix[ArcCos[r / Norm[p - cen]]].
        (p - cen)]
  ];
  Graphics[
   {
    EdgeForm[Black], Orange,
    {Opacity[.3], Disk[], Disk[cen, r]},
    Black, Line[{p1, p2}],
    If[Norm[p] > 1, {Blue, Line[{t2, p, t1}], Yellow,
      Disk[t1, .06], Disk[t2, .06]}, {}],
    Red, Disk[p, .06]
   },
   PlotRange → 3, ImageSize → {400, 400}
  ]
 ],

 {{cen, {.5, .4}}, {-3, -3}, {3, 3}, Locator},
 {{r, 1, "radius of circle"}, 0, 3, Appearance → "Labeled"},
 {{rp, .3, "position of red point on radical axis"},
  0, 1, Appearance → "Labeled"}
]
```



◻ Crossing Arbelos and 3D Arbelos

The following `Manipulate` illustrates two generalizations.

```
Manipulate[
 Module[{m, ri, h, k, r, li, t1, t2},
  m = {(1 + a) (1 - b), 2 √(a b (a - 1) (b - 1))}/ (1 - a b);
  ri = (1 - a) (1 - b) / (1 - a b);
  h = {(1 + a) (1 - b), 2 √(a (a - 1) (b - 1))}/ (2 - a - b);
  k = {(3 - a) (1 - b), 2 √(b (a - 1) (b - 1))}/ (2 - a - b);
```

```
r = (1 - a) (1 - b) / (2 - a - b);
li = 2 (1 - b) / (2 - a - b);
t1 = Chop@If[2 a < 2 - 2 b, 0, ArcCos[(li - a) / a]];
t2 = Chop@If[2 a < 2 - 2 b, 0, ArcCos[(2 - li - b) / b]];
If[
 threeD,
 Graphics3D[
  {
   Sphere[{a, 0, 0}, a], Sphere[{2 - b, 0, 0}, b],
   Sphere[{1, 0, 0}, 1], Opacity[.7],
   Cylinder[{{1, 0, 0}, {1, -2, 0}}, 1],
   If[
    tw,
    {Sphere[{First[h], Last[h], 0}, r],
     Sphere[{First[k], Last[k], 0}, r],
     Cylinder[{{li, 0, 0}, {li + .001, 0, 0}},
       .99 √(li (2 - li))]},
    {}
   ]
  }, Boxed → False, ViewPoint → {0, 0, ∞},
  PlotRange → {{-.1, 2}, {0, 1}, {0, -2}},
  ImageSize → {425, 210}],
 Graphics[{
   ColorData[2, 5], Disk[{1, 0}, 1, {0, π}],
   White, Disk[{a, 0}, a, {0, π}],
   Disk[{2 - b, 0}, b, {0, π}],
   Black,
   Circle[{a, 0}, a, {t1, π}],
   Circle[{2 - b, 0}, b, {0, π - t2}],
   Circle[{1, 0}, 1, {0, π}],
   If[ic, {ColorData[2, 9], EdgeForm[Black], Disk[m, ri]},
    {}],
   If[
    tw,
    {ColorData[2, 6], EdgeForm[Black], Disk[h, r],
     Disk[k, r]},
    {}
   ],
   Line[{{li, 0}, {li, √(li (2 - li))}}],
   Arrow[{{a, 0}, {a, 0} + a {-1, 1} 0.7071}],
```

```
      Arrow[{{2 - b, 0}, {2 - b, 0} + b {1, 1} 0.7071}],
      Point[{a, 0}], Point[{2 - b, 0}],
      If[a + b < 1, Line[{{2 a, 0}, {2 - 2 b, 0}}], {}],
      Style[
       {
        Text[Style["a", Italic],
         {a, -0.05} + 0.5 a {-1, 1} 0.7071],
        Text[Style["b", Italic],
         {2 - b, -0.05} + 0.5 b {1, 1} 0.7071]
       }, 12]
     }, PlotRange → {{-.1, 2.1}, {-.08, 1.1}},
     ImageSize → {425, 210}]
   ]
  ],

  Row[{
    Spacer[40],
    Control[{{a, .3001, Style["a", Italic]}, 0, .999,
      .0001, Appearance -> "Labeled", ImageSize → Small}],
    Spacer[20],
    Control[{{b, .7, Style["b", Italic]}, .999, 0,
      .0001, Appearance → "Labeled", ImageSize → Small}]
   }],
  Row[{
    Spacer[100],
    Control[{{threeD, False, "3D version"},
      {True, False}}],
    Spacer[7],
    Control[{{ic, False, "show incircle"}, {True, False},
      Enabled :→ Not@threeD}],
    Spacer[7],
    Control[{{tw, False, "show twins"}, {True, False}}]
   }]
 ]
```



Property 28

The inscribed circles tangent to the radical axis of the side arcs and the top arc and either of the arcs of the generalized arbelos have the same radius.

Let g be the length of the *gap* between the bases (so that the diameter of the top arc is $2a + g + 2b$) and let v be the abscissa of the intersection of the radical axis with the x axis, assuming the origin is at the leftmost point of the arbelos [10].

Theorem 5

If circles $\sigma_1 = \odot(c_1, r_1)$ *and* $\sigma_2 = \odot(c_2, r_2)$ *do not intersect, their radical axis meets the segment* $c_1\, c_2$ *in the point* p *such that* $|p - c_1|^2 - |p - c_2|^2 = r_1^2 - r_2^2$.

With the help of this theorem, we compute the value of v.

```
Clear[a, b, g, v];
Solve[(v - a)^2 - (2 a + g + b - v)^2 == a^2 - b^2, v]
```

$$\left\{\left\{v \to \frac{(2\,a + g)\,(2\,a + 2\,b + g)}{2\,(a + b + g)}\right\}\right\}$$

We can assume without loss of generality that $0 < a < 1$, $0 < b < 1$, and $-2 < g < 2$ (g can be negative). Let the inscribed circles be $\odot(\alpha, \rho)$ and $\odot(\beta, \sigma)$. The values of these parameters are obtained as follows.

```
Clear[a, b, g, v, α, ρ, β, σ];
Module[{v, x1, y1, x2, y2, r1, r2},
 v = ((2 a + g) (2 a + 2 b + g))/(2 (a + b + g));
 Union[
  Flatten[
   Simplify[
    {{α → {x1, y1}, ρ → r1}, {β → {x2, y2}, σ → r2}} /.
     Solve[
      {
       dSq[{x1, y1}, {a, 0}] == (a + r1)^2,
       dSq[{x1, y1}, {(2 a + 2 b + g) / 2, 0}] ==
        ((2 a + 2 b + g) / 2 - r1)^2,
       dSq[{x2, y2}, {2 a + b + g, 0}] == (b + r2)^2,
       dSq[{x2, y2}, {(2 a + 2 b + g) / 2, 0}] ==
        ((2 a + 2 b + g) / 2 - r2)^2, r1 == v - x1, r2 == x2 - v
      },
      {x1, y1, x2, y2, r1, r2}
     ], (0 < a < 1) ∧ (0 < b < 1) ∧ (-2 < g < 2)],
   1
  ]
 ]
]
```

$$\left\{\left\{\alpha \to \left\{\frac{(2\,a+g)\,(4\,a+2\,b+g)}{4\,(a+b+g)},\ -\frac{\sqrt{a\,(2\,a+g)}\,\sqrt{(2\,b+g)\,(2\,a+2\,b+g)}}{\sqrt{2}\,\text{Abs}[a+b+g]}\right\},\ \rho \to \frac{(2\,a+g)\,(2\,b+g)}{4\,(a+b+g)}\right\},\ \left\{\alpha \to \left\{\frac{(2\,a+g)\,(4\,a+2\,b+g)}{4\,(a+b+g)},\ \frac{\sqrt{a\,(2\,a+g)}\,\sqrt{(2\,b+g)\,(2\,a+2\,b+g)}}{\sqrt{2}\,\text{Abs}[a+b+g]}\right\},\ \rho \to \frac{(2\,a+g)\,(2\,b+g)}{4\,(a+b+g)}\right\},\ \left\{\beta \to \left\{\frac{(2\,a+g)\,(4\,a+6\,b+3\,g)}{4\,(a+b+g)},\ -\frac{\sqrt{b\,(2\,b+g)}\,\sqrt{(2\,a+g)\,(2\,a+2\,b+g)}}{\sqrt{2}\,\text{Abs}[a+b+g]}\right\},\ \sigma \to \frac{(2\,a+g)\,(2\,b+g)}{4\,(a+b+g)}\right\},\ \left\{\beta \to \left\{\frac{(2\,a+g)\,(4\,a+6\,b+3\,g)}{4\,(a+b+g)},\ \frac{\sqrt{b\,(2\,b+g)}\,\sqrt{(2\,a+g)\,(2\,a+2\,b+g)}}{\sqrt{2}\,\text{Abs}[a+b+g]}\right\},\ \sigma \to \frac{(2\,a+g)\,(2\,b+g)}{4\,(a+b+g)}\right\}\right\}$$

Then, although some centers can be disregarded, the radius is the same in all cases.

Proof without Words

Finally, here are three more properties of the arbelos. See if you can guess what property is involved by experimenting with the controls [24, 25].

This first `Manipulate` lets you move the side arcs in a systematic way.

```
Manipulate[
 Module[
  {h, k, x, y, u, v},
  If[b > Sqrt[1 - a^2], b = Sqrt[1 - a^2] - .001];
  {h} = x /. NSolve[Norm[Normalize[{x, b}] - {x, b}] == a - x, x];
  {k} = x /. NSolve[Norm[Normalize[{x, b}] - {x, b}] == x - a, x];
  y = (a + 1) b / (h + 1);
  {u, v} = Normalize /@ {{h, b}, {k, b}};
  Graphics[
   {
    EdgeForm[Black],
    ColorData[2, 5], Disk[{0, 0}, 1],
    ColorData[2, 6], Disk[{h, b}, a - h], Disk[{k, b}, k - a],
    Black, Disk[#, .02] & /@ {{h, b}, {k, b}, {a, y}, u, v},
    Line[{{-1, 0}, {1, 0}}],
    Line[{{a, 0}, {a, Sqrt[1 - a^2]}}],
    {Dashed, Line[{{-1, 0}, {a, y}, {1, 0}}],
     Line[{u, v, {0, 0}, u}]},
    Red, Disk[{a, b}, .02]
   }, PlotRange → 1.1, ImageSize → {400, 400}
  ]
 ],

 {{a, .3, "vertical line"}, -.999, .999,
  Appearance → "Labeled"},
 {{b, .2, "red point"}, 0, 1, Appearance → "Labeled"}
]
```


vertical line
0.3
red point
0.2

This second `Manipulate` lets you rotate a line around the point of tangency of the side arcs.

```
Manipulate[
 Module[
  {h, u, α, k, w, v, z},
  h = {Cos[t], Sin[t]};
  u = {2 a - 1, 0};
  α = (1 - u.u) / (1 + u.(u - 2 h));
  k = u - α (h - u);
  w = k + ({a - 1, 0} - k).Normalize[{2 a - 1, 0} - k]
     Normalize[{2 a - 1, 0} - k];
  v = 2 w - {2 a - 1, 0};
  z = k + h - v;
  If[Norm[k - v] > Norm[h - v], {v, z} = {z, v}];
  Graphics[
   {
    Blue, Line[{v, k}], Line[{z, h}],
    EdgeForm[Black],
    ColorData[2, 5], Disk[{0, 0}, 1],
    ColorData[2, 6], Disk[{a - 1, 0}, a], Disk[{a, 0}, 1 - a],
    Black, {Dashed, Line[{h, k}]},
    Red, Disk[#, .015] & /@ {h, k, v, z}
   }, PlotRange → 1.1, ImageSize → {400, 400}]],

 {{a, .6, Row[{"radius ", Style["a", Italic],
     " of left side arc"}]}, 0.001, 0.999, .001,
  Appearance → "Labeled"},
 {{t, 1, "rotate line"}, 0, 2 π, Appearance → "Labeled"}
]
```


radius *a* of left side arc
0.6
rotate line
1

Finally, the third `Manipulate` shows an infinite family of twins.

```
Manipulate[
 Module[{c, cx, cy, α, t1, t1x, t1y, t2, r2, h},
  {cx, cy} = c = {1 + a, 0} + (1 - a) {Cos[t], Sin[t]};
  α = ArcCos[a / Norm[c - {a, 0}]];
  t1 = {t1x, t1y} =
    {a, 0} + a Normalize[RotationMatrix[α].(c - {a, 0})];
  t2 = {a, 0} + a Normalize[RotationMatrix[-α].(c - {a, 0})];
  r2 = a (1 - a);
  h = {a, 0} + (1 - Norm[c - t1] / Abs[-cy t1x + a (cy - t1y) + cx t1y] Abs[r2])
     (c - {a, 0});
  Graphics[
   {
    EdgeForm[Black], ColorData[2, 5],
    Disk[{1, 0}, 1, {0, π}],
    White, Disk[{a, 0}, a], Disk[{1 + a, 0}, 1 - a, {0, π}],
    ColorData[2, 6], Disk[h, r2],
    Black,
    Circle[{1, 0}, 1, {0, π}],
    Circle[{a, 0}, a, {0, π}],
    Circle[{1 + a, 0}, 1 - a, {0, π}],
    Line[{t1, c, t2}],
    EdgeForm[Black], Red, Map[Disk[#, .015] &, {c, t1, t2}]
   }, PlotRange → {{-.1, 2.1}, {-1, 1}}, ImageMargins → 10,
   ImageSize → {400, 400}]],

 {{a, .6, Row[{"radius ", Style["a", Italic],
     " of left side arc"}]}, 0.001, .999, .001,
  Appearance → "Labeled"},
 {{t, 1.5, "intersection of tangents"}, 0, π - .001,
  Appearance → "Labeled"}
]
```


radius *a* of left side arc
0.6
intersection of tangents
1.5

References

[1] F. van Lamoen. "Online Catalogue of Archimedean Circles." (Jan 22, 2014) home.planet.nl/~lamoen/wiskunde/arbelos/Catalogue.htm.

[2] S. Garcia Diethelm. "Planar Stress Rotation" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/PlanarStressRotation.

[3] C. W. Dodge, T. Schoch, P. Y. Woo, and P. Yiu, "Those Ubiquitous Archimedean Circles," *Mathematica*l Magazine, **72**(3), 1999 pp. 202–213. www.jstor.org/stable/2690883.

[4] H. P. Boas, "Reflection on the Arbelos," *American Mathematical Monthly*, **113**(3), 2006 pp. 236–249.

[5] H. D. Dörrie, *100 Great Problems of Elementary Mathematics: Their History and Solution* (D. Antin, trans.), New York: Dover Publications, 1965.

[6] J. Rangel-Mondragón. "Recursive Exercises II: A Paradox" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/RecursiveExercisesIIAParadox.

[7] R. B. Nelsen, "Proof without Words: The Area of an Arbelos," *Mathematics Magazine*, **75**(2), 2002 p. 144.

[8] A. Gadalla. "Area of the Arbelos" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/AreaOfTheArbelos.

[9] J. Rangel-Mondragón, "Selected Themes in Computational Non-Euclidean Geometry. Part 1. Basic Properties of Inversive Geometry," *The Mathematica Journal*, 2013. www.mathematica-journal.com/2013/07/selected-themes-in-computational-non-euclidean-geometry-part-1.

[10] D. Pedoe, *Geometry: A Comprehensive Course*, New York: Dover, 1970.

[11] M. Schreiber. "Orthogonal Circle Inversion" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/OrthogonalCircleInversion.

[12] M. G. Welch, "The Arbelos," Master's thesis, Department of Mathematics, University of Kansas, 1949.

[13] L. Bankoff, "The Marvelous Arbelos," *The Lighter Side of Mathematics* (R. K. Guy and R. E. Woodrow, eds.), Washington, DC: Mathematical Association of America, 1994.

[14] G. L. Alexanderson, "A Conversation with Leon Bankoff," *The College Mathematics Journal*, **23**(2),1992 pp. 98–117.

[15] S. Kabai. "Tangent Circle and Arbelos" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/TangentCircleAndArbelos.

[16] G. Markowsky and C. Wolfram. "Theorem of the Owl's Eyes" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/TheoremOfTheOwlsEyes.

[17] P. Y. Woo, "Simple Constructions of the Incircle of an Arbelos," *Forum Geometricorum*, **1**, 2001 pp. 133–136. forumgeom.fau.edu/FG2001volume1/FG200119.pdf.

[18] B. Alpert. "Archimedes' Twin Circles in an Arbelos" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/ArchimedesTwinCirclesInAnArbelos.

[19] J. Rangel-Mondragón. "Twins of Arbelos and Circles of a Triangle" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/TwinsOfArbelosAndCirclesOfATriangle.

[20] H. Okumura, "More on Twin Circles of the Skewed Arbelos," *Forum Geometricorum*, **11**, 2011 pp. 139–144. forumgeom.fau.edu/FG2011volume11/FG201114.pdf.

[21] E. W. Weisstein. "Point-Line Distance—2-Dimensional" from Wolfram *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/Point-LineDistance2-Dimensional.html.

[22] L. Bankoff, "Are the Twin Circles of Archimedes Really Twins?," *Mathematics Magazine*, **47**(4), 1974 pp. 214–218.

[23] F. Power, "Some More Archimedean Circles in the Arbelos," *Forum Geometricorum*, **5**, 2005 pp. 133–134. forumgeom.fau.edu/FG2005volume5/FG200517.pdf.

[24] A. V. Akopyan, *Geometry in Figures*, CreateSpace Independent Publishing Platform, 2011.

[25] H. Okumura and M. Watanabe, "Characterizations of an Infinite Set of Archimedean Circles," *Forum Geometricorum*, **7**, 2007 pp. 121–123. forumgeom.fau.edu/FG2007volume7/FG200716.pdf.

J. Rangel-Mondragón, "The Arbelos," *The Mathematica Journal*, 2014. dx.doi.org/doi:10.3888/tmj.16-5.

About the Author

Jaime Rangel-Mondragón received M.Sc. and Ph.D. degrees in applied mathematics and computation from the University College of North Wales in Bangor, UK. He has been a visiting scholar at Wolfram Research, Inc. and has held positions in the Faculty of Informatics at UCNW, the College of Mexico, the Center for Research and Advanced Studies, the Monterrey Institute of Technology, the Queretaro Institute of Technology, and the University of Queretaro in Mexico, where he is presently a member of the Faculty of Informatics. His current research includes combinatorics, the theory of computing, computational geometry, urban traffic, and recreational mathematics.

Jaime Rangel-Mondragón
UAQ, Facultad de Informatica
Queretaro, Qro. Mexico
jrangelmondragon@gmail.com